
First, more on LLVM:

Thursday, October 18, 12



Branching:
Returning from functions:
 ret i8 %x 

Unconditional branching:
 br label %L
 ...
 L: 
 ...

Conditional branching:
 br i1 %y, label %T, label %F
 T: ret i32 888
 F: ret i32 999

Label references 
start with a %

Note: two 
destinations!

Basic Blocks:
LLVM instruction sequences are implicitly structured 
as a graph of basic blocks.

A basic block is a sequence of instructions that 
starts with a leader and ends with a terminator.
n ret and br (and some exotic instructions) are terminators.

n The first instruction of a function, as well as the successor 
to a terminator, are leaders.

Labels may only be attached to leaders.

Labels = names of basic blocks.

(Leaders with no labels receive labels from the 
series of anonymous names: %0, %1, ...)

Basic Blocks Example:
C/Java: void f(int x) {
       if (x % 7 == 0)
           g(x);
       else
           h(x);
  }

LLVM:  define @f(i32 %x) {
     init: %1 = srem i32 %x, 7
   %2 = icmp eq i32 %1, 0
   br i1 %2, label %T, label %F

        T: call void @g(i32 %x)
   br label %exit

        F:  call void @h(i32 %x)
   br label %exit

     exit: ret void
  }

init

T F

exit

Thursday, October 18, 12



SSA
So far, LLVM appears like a strange C/assembler 
hybrid

However, the analysis & optimization power of LLVM 
comes from a restriction on the way local variables 
may be assigned:

n In LLVM, each local variable must be assigned a 
value exactly one per function!

This restriction is called Static Single Assignment 
form (SSA), and has proven very successful in 
optimizing compiler developments.

SSA
The benefit of SSA is that for each use of a local 
variable, there is a unique place in the basic block 
graph where it is defined.

The SSA restriction renders LLVM code some 
properties similar to purely functional languages — 
the freedom to share and reorder computations for 
better hardware utilization being one.

Load and store operations still complicate the 
picture, though, as does calls to unknown functions.

Generating SSA
To generate LLVM code from arbitrary MiniJava 
instruction sequences that may mutate variables, one 
must increase a version number each time a variable 
is assigned, and keep track of which version that is 
referenced in each variable expression.
For straight sequential code, the version referenced is 
just that of the last assignment:
 a = b + c;                   %a.1 = add i32 %b.0 %c.0
 b = a - c;                   %b.1 = sub i32 %a.1 %c.0
 c = b + c;                   %c.1 = add i32 %b.1 %c.0

However, for code that contains branches and loops, 
the SSA restriction becomes a little more involved.

SSA and branches
Consider the following example:

 x = f(7);   x0 = f(7);
 if (x > 0)   if (x0 > 0)
     x = x - 1;       x1 = x0 - 1;
 else    else
     x = x + 1;       x2 = x0 + 1;
 return x;   return x?;

There are three assignements to x, but for each run of the 
example, one of the assignments will be skipped.
Which version of x should be returned? (Informal answer: 
the one "not skipped"...)

SSA and loops
Consider the following example:

 x = f(7);   x0 = f(7);
 while (x > 0)   while (x? > 0)
     x = x - 7;       x1 = x? - 7;

There are only two assignements to x, but there will be an 
unknown number of x versions at run-time.
Which version should be used when computing x > 0 and 
x - 7? (Informal answer: the "last" one...)

The SSA solution: phi nodes!
The problem occurs when a basic block can be reached via 
different basic block paths, and some variable x receives 
different versions along these paths.
Solution: insert a purely bureauchratic operation phi, that 
joins different versions of x!

L1:
    %x.1 = ...

L2:
    %x.2 = ...

L3:
    %x.3 = phi i32 [ %x.1, %L1 ], [ %x.2, %L2 ]
    ret %x.3

The Loop Example:

L2:  

   %x.2 = sub i32 %x.1, 1
   br label %L1

L1:  

   %x.1 = phi i32 [ %x.0, %L0 ], [ %x.2, %L2 ]
   %test = icmp sge i32 %x.1, 0
   br i1 %test, label %L2, label %L3

L3:  

    ...

L0:  

   %x.0 = call @f (i32 7)
   br label %L1

An Slightly Bigger Example:

int max(int n, int a[]) {
   int x = 0;
   int i = 0;
   while (i < n) {
      if (a[i] > x)
         x = a[i];
      i = i + 1;
   }
   return x;
}

define i32 @max(i32 %n, i32* %a) {
  %1 = icmp sgt i32 %n, 0
  br i1 %1, label %L1, label %L2
L1:
  %i0 = phi i32 [ %i1, %L1 ], [ 0, %0 ]
  %x0 = phi i32 [ %x1, %L1 ], [ 0, %0 ]
  %2 = getelementptr i32* %a, i32 %i0
  %3 = load i32* %2
  %4 = icmp sgt i32 %3, %x0
  %x1 = select i1 %4, i32 %3, i32 %x0
  %i1 = add i32 %i0, 1
  %5 = icmp eq i32 %i1, %n
  br i1 %5, label %L2, label %L1
L2:
  %x2 = phi i32 [ 0, %0 ], [ %x1, %L1 ]
  ret i32 %x2
}

Summary:
LLVM offers an intermediate representation and a modular 
infrastructure for code optimization.

LLVM code can be represented as text, compressed 
binaries, or an AST in C++.

LLVM is typed, and offers the usual set of arithmetic, 
bitwise and logical operations.

Memory is accessed via load and store, offset calculations 
are done by getelementptr.

Function bodies reveal an explicit basic block structure.

Variable assignments follow the SSA restriction.

The phi operator joins variables from multiple basic blocks.
14

Generating LLVM: 

Our simple compiler:

Without an intermediate representation, we use a 
form of “syntax-directed code generation” to 
produce IA32 code directly from the validated 
abstract syntax.

Reasonable output, reasonably low cost.

Lexer Parser Static 
Analysis

IA32 Code 
Generator

An Optimizing Compiler:

For better output (but higher cost) we 
will generate intermediate code instead, 
for further processing by optimizers.

Lexer Parser Static 
Analysis

Intermed. 
Code 

Generator

Target Code 
Generator

Optimizer(s)

Let's see if we can modify our IA32 code generator 
to produce LLVM code in a syntax-directed manner.

Generating LLVM code:

Recall our previous compilation scheme Er[e](sp), 
that generates assembly code for e, placing the 
result in register sp.
Parameter sp is an indicator of the current register 
pressure, as it only increases while another 
temporary value needs storeage.
With LLVM we don't need to reuse registers to keep 
the register pressure as low as possible — in fact, we 
cannot, due to the SSA restriction!
Instead we will assume the existence of an infinite 
name-supply for inventing new variable names.

An LLVM compilation scheme:
Let Ellvm[e](x) stand for the LLVM code that evaluates e 
and leaves the result in %x.

For example:

  Ellvm[e1+e2](x)  = Ellvm[e1](x1)  (x1 is new)

     Ellvm[e2](x2)  (x2 is new)

     %x = add i32 %x1,%x2

Note: since %n is a valid register name for any positive 
n, the name supply of our compiler can be just a global 
integer variable that is read and incremented each time 
a new name is requested.

19

More schemes:
Continuing this idea, we would prefer to translate variable 
and constant expressions as follows:

  Ellvm[v](x)     = %x = %v

  Ellvm[n](x)     = %x = n
Unfortunately, LLVM disallows copy instructions of the 
form above (it's not entirely clear why).
We could comply by implementing copy propagation...
But a much simpler solution is to dress up a leaf 
expression as a harmless operation! Here's a suggestion:
 Ellvm[v](x)     = %x = select i1 true, tv %v, tv %v

  Ellvm[n](x)     = %x = select i1 true, i32 n, i32 n
20

Thursday, October 18, 12



Compiling conditionals:
No special schemes for conditional expressions are 
necessary — let the LLVM optimizers do the job!

  Ellvm[e1>e2](x)  = Ellvm[e1](x1)  (x1 is new)

     Ellvm[e2](x2)  (x2 is new)

     %x = icmp sgt i1 %x1,%x2

Compiling a conditional statement:

 Sllvm[while (e) s]  =        br label %L2

     L1:  Sllvm[s]
     L2:  Ellvm[e](x) (x is new)

             br i1 %x, label %L1, label %L3
     L3:   

21

LLVM and Mutation...
Due to the SSA restriction of LLVM, we cannot 
simply translate assignments in MiniJava as
 Sllvm[v = e]     = Ellvm[e](v)

What SSA demands is that we
n invent a new name for each updated version of v, 

and use that name for subsequent accesses.
n insert phi operations whenever the basic block graph 

makes access to more than one version possible.

Unfortunately, the general solution to this 
problem requires an analysis that traverses the 
basic block graph (which we don't want to build...)

22

(Inserting the phi...)

0

A flow graph
(of basic blocks)

Its dominance 
("must pass") tree

The dominance frontier: DFRONT(n) = the set of "first" 
nodes reachable from n but not dominated by n.

If node n assigns to a
variable x, an operation
    xnew = phi x
that joins x-versions xs
must be inserted into
each node in DFRONT(n)

LLVM and Mutation:
Building flow graphs and computing dominance frontiers is 
much more work than doing syntax-directed translation

In fact, computing DFRONT is an example of a so called 
data-flow analysis, a common algorithmic pattern in 
optimizing code transformations.

It would be quite ironic if LLVM would force us to implement 
such machinery, just in order to leverage on exising 
transformations that already use similar techniques!

Fortunately, we can avoid the complexities of SSA 
altogether, by storing all our MiniJava variables on the stack 
(memory accesses don't have the SSA restriction)

The efficiency lost can be recovered later, by an LLVM 
transformation that replaces stack variables with registers!

LLVM with Stack Allocation:
Compiling a local variable declaration:

 Sllvm[t v; s]  = %v = alloca Tllvm[t]
     Sllvm[s]

Compiling a local variable reference:

  Ellvm[v](x)  =  %x = load tv* %v

Compiling a local variable assignment:

 Sllvm[v = e]  = Ellvm[v](x)  (x is new)

     store tv %x, tv* %v

25

Transforming Types:
Since LLVM is typed, types need translation as well:

 Tllvm[ int ]      = i32

  Tllvm[ boolean ]   = i1

  Tllvm[ int[] ]     = i32*

  Tllvm[ C ]      = %C*

This assumes a declaration class C { t1 v1 ... tn vn; mdecls } 
is matched by a corresponding LLVM declaration

  %C = type { Tllvm[ t1 ], ..., Tllvm[ tn ] }

(Note: correctly translated types are also needed at 
several places where our schemes just say tv.)

26

Transforming Methods:
In general, transforming methods into LLVM functions 
works just like transformation into IA-32 subroutines.
That means:
n Name mangling 
n Insertion of the extra parameter this
n Accessing instance variables as offsets from this

A method header t m ( t1 f1, ..., tn fn ) for class C may in 
principle be translated into

    Tllvm[ t ] @m ( %C* %this, Tllvm[ t1 ] %f1, ..., Tllvm[ tn ] %fn )

However, this is assuming that formal parameters f1...fn 
are not updated. Those that are must be turned into local 
variables, that get initialized by formals with new names.

27

Optimizing LLVM: 

The 'opt' Tool:
The tool opt is a generic optimization driver which 
reads and writes LLVM files, performing 
configurable sequences of optimizations in 
between.
LLVM code in text format is stored in .ll files.
To run optimization algorithm X on file.ll and have 
the result printed on stdout, type
 > opt -S -X file.ll
To see which optimization algorithms that are 
currently installed, type
 > opt -help

29

With 'opt -mem2reg'

L0:
  %x = alloca i32
  %0 = icmp sge i32 %n, 0
  br i1 %0, label %L1, label %L2
L1:
  %1 = sub i32 %n, 1
  store i32 %1, i32* %x
  br label %L3
L2:
  %2 = add i32 %n, 1
  store i32 %2, i32* %x
  br label %L3
L3:
  %3 = load i32* %x
  ret i32 %3

L0:
  %0 = icmp sge i32 %n, 0
  br i1 %0, label %L1, label %L2

L1:
  %1 = sub i32 %n, 1
  br label %L3

L2:
  %2 = add i32 %n, 1
  br label %L3

L3:
  %x.0 = phi i32 [ %1, %L1 ], [ %2, %L2 ]
  ret i32 %x.0

With 'opt -analyze -domfrontier'

L0:
  %x = alloca i32
  %0 = icmp sge i32 %n, 0
  br i1 %0, label %L1, label %L2
L1:
  %1 = sub i32 %n, 1
  store i32 %1, i32* %x
  br label %L3
L2:
  %2 = add i32 %n, 1
  store i32 %2, i32* %x
  br label %L3
L3:
  %3 = load i32* %x
  ret i32 %3

  DomFrontier for BB %L0 is: 
  DomFrontier for BB %L1 is:  %L3
  DomFrontier for BB %L2 is:  %L3
  DomFrontier for BB %L3 is: 

With 'opt -instsimplify'

L0:
  %0 = select i1 true, i32 %n, i32 %n
  ret i32 %0

L1:
  %0 = add i32 %n, 0
  ret i32 %0

L2:
  %0 = and i32 %n, -1
  ret i32 %0

L0:
  ret i32 %n

L1:
  ret i32 %n

L2:
  ret i32 %n

With 'opt -inline'

define i32 @f(i32 %k, i32 %l) {
  %kl = mul i32 %k, %l
  ret i32 %kl
}

L0:
  %0 = call i32 @f (i32 %n, i32 2)
  %1 = add i32 %0, 1
  %2 = call i32 @f (i32 %n, i32 2)
  %3 = add i32 %2, 1
  %4 = add i32 %1, %3
  ret i32 %4

L0:
  %kl.0 = mul i32 %n, 2
  %0 = add i32 %kl.0, 1
  %kl.1 = mul i32 %n, 2
  %1 = add i32 %kl.1, 1
  %2 = add i32 %0, %1
  ret i32 %2

With 'opt -early-cse'

L0:
  %kl.0 = mul i32 %n, 2
  %0 = add i32 %kl.0, 1
  %kl.1 = mul i32 %n, 2
  %1 = add i32 %kl.1, 1
  %2 = add i32 %0, %1
  ret i32 %2 L0:

  %kl.0 = mul i32 %n, 2
  %0 = add i32 %kl.0, 1
  %1 = add i32 %0, %0
  ret i32 %1

With 'opt -inline' (2)

35

define i32 @f(i32 %k, i32 %l) {
  %kl = mul i32 %k, %l
  ret i32 %kl
}

L0:
  %0 = call i32 @f (i32 %n, i32 2)
  %1 = add i32 %0, 1
  %2 = call i32 @f (i32 %n, i32 2)
  %3 = add i32 1, %2
  %4 = add i32 %1, %3
  ret i32 %4

L0:
  %kl.0 = mul i32 %n, 2
  %0 = add i32 %kl.0, 1
  %kl.1 = mul i32 %n, 2
  %1 = add i32 1, %kl.1
  %2 = add i32 %0, %1
  ret i32 %2

With 'opt -early-cse' (2)

L0:
  %kl.0 = mul i32 %n, 2
  %0 = add i32 %kl.0, 1
  %kl.1 = mul i32 %n, 2
  %1 = add i32 1, %kl.1
  %2 = add i32 %0, %1
  ret i32 %2 L0:

  %kl.0 = mul i32 %n, 2
  %0 = add i32 %kl.0, 1
  %1 = add i32 1, %kl.0
  %2 = add i32 %0, %1
  ret i32 %2

Thursday, October 18, 12



With 'opt -reassociate'

L0:
  %kl.0 = mul i32 %n, 2
  %0 = add i32 %kl.0, 1
  %1 = add i32 1, %kl.0
  %2 = add i32 %0, %1
  ret i32 %2

L0:
  %factor = mul i32 %n, 4
  %0 = add i32 %factor, 2
  ret i32 %0

Induction Variables:
Consider the following loop:
 for (i = 7; i < N; i++)
     x += a[i*5];

Each turn of the loop must perform a multiplication.
However, each time the counter i increases by 1, the 
array index increases by 5.
Thus, the following loop would be equivalent:
 int *iv = a + 35;
 for (i < 7; i < N; i++) {
     x += *iv;
     iv = iv + 5;
 }

A variable like iv above is called an induction variable.
Could the transformation above be automated?

38

Yes: 'opt -indvars -loop-reduce'

L0:
  br label %L1
  
L1:
  %i0 = phi i32 [ 7, %L0 ], [ %i1, %L1 ]
  %x0 = phi i32 [ 0, %L0 ], [ %x1, %L1 ]

  %j = mul i32 %i0, 5
  %adr = getelementptr i32* %a, i32 %j
  %val = load i32* %adr
  
  %i1 = add i32 %i0, 1
  %x1 = add i32 %x0, %val

  %test = icmp slt i32 %i1, 1000
  br i1 %test, label %L1, label %L2

L2:
  ret i32 %x1

L0:
  %iv0 = getelementptr i32* %a, i32 35
  br label %L1

L1:
  %iv = phi i32* [ %iv0, %L0 ], [ %iv1, %L1 ]
  %i0 = phi i32 [ 7, %L0 ], [ %i1, %L1 ]
  %x0 = phi i32 [ 0, %L0 ], [ %x1, %L1 ]

  %v = load i32* %iv
  %x1 = add i32 %x0, %v
  %i1 = add i32 %i0, 5
  %iv1 = getelementptr i32* %iv, i32 5

  %test = icmp slt i32 %i1, 1000
  br i1 %test, label %L1, label %L2

L2:
  ret i32 %x1

Summary:
LLVM optimization can be explored using the 
opt tool.
Examples of optimizations:
n instruction simplification
n replacing stack storage with registers
n common subexpression elimination
n arithmetic reassociation
n inlining
n identification of induction variables

Many more optimizations exist, use opt -help to 
explore!

